		Synchrones ROM "Write before Read"	11
Digital Design		Synchrone Logik	12
		Synchronisation & Entprellung	13
Zusammenfassung		Metastabilität	13
		Reset Synchronisierung	13
Joel von Rotz & Andreas Ming / Quelldate	eien	Entprellen	13
		durch Blanking	13
Table of contents		durch Unterabtastung	14
Table of contents —		Drehgeber-Signale (Quadratur-Signale)	14
FPGA	2		
Technologie	2	Finite State Machines (FSM)	14
Layout	2	FSM-Typ: Mealy	14
Routing Ressourcen	2	Parasitäre Zustände	15
	_	State Encoding	15
Integrated Logic Analyzer (ILA)	3	Binär	15
Kurzanleitnung	3	One-Hot	15
VHDL	3	Goldene Regeln der (FSM) Implementierung	15
Einordnung VHDL als HDL	4	Memoryless Process (kombinatorische Logik)	15
Entwicklung	4	Memorizing Process (sequentielle Logik)	15
Designflow	4	momentaling is recess (sequentilene Legin)	
Simulation	4	Fort / and Claitheanne Arithmetic	1.
Event Driven Simulation	5	Fest-/ und Gleitkomma-Arithmetik	15
Signale & Variablen	5	Festkomma	15
Sprachelemente	5 5	Addition	16
Signalzuweisung / Treiber <=	5	Multiplikation	16
Kommentare	5	Gleitkomma	16
Komponenten	5		
.vhd-Dateistruktur	5	PWM-D/A	17
Entity	5		
Generics	6	Packages	17
Loop & Generate Statements	6	Libraries & Use Clause	18
Datentypen	7	Liste von Packages	18
Integer	7	Synthetisierbare Bibliotheken	18
Aufzählungstypen	7 7	ieee.std_logic_1164	18
Array	7	Nicht-Synthetisierbare Bibliotheken	18
Record	8	,	
Signed/Unsigned (IEEE 1076.3)	8	Vorlagen	18
IEEE std_logic_1164	9	-	
Real (Simulationstype)	9	Positive Getriggertes D-FlipFlop	18
Time (Simulationstyp)	9	Mit asynchronem Reset	18
Prozesse	9	Ohne Reset	18
wait (nur für Simulation)	10	Finite State Machine	19
Process Statement	10	Mealy	19
Sequential Statements	10	Moore	19
Schaltungssynthese	10	Synchronisation	20
Synthese	10	einfach	20
Implementation	10	mit Flankenerkennung	20
Bitstream	10	Entprellen	20
Speichermodellierung	11	durch Blanking & Unterabtastung	20
kombinatorisch ROM	11	durch Unterabtastung	21

FPGA ----

Warum FPGA?

FPGA steht für **F***ield* **P***rogrammable* **G***ate* **A***rray* und ist die am weitesten verbreite Art von "programmierbarer" Logik.


Gegenüber einem anwendungsspezifischen Chip (ASIC) bieten FPGA:

- + höherer Flexibilität ; kürzere Entwicklungszeit ; geringer Entwicklungskosten
- im höhere Frequenzbereich \to Um mitzuhalten sind Synchronisationen nötig, was zu Signal-Latenzen führt. Verfügt über:
 - ullet Parrallelität o beschleunigte Verarbeitung
 - Flexible Zuweisungen von Signalen & Pin-Funktionalitäten
 - Deterministische Durchlaufzeiten von Signalen (z.B. 0cc, 2cc)
 - Können mehrere Prozessoren beinhalten \rightarrow erhöhter Integrationsgrad
 - Custom Peripherie für Mikrocontroller


! FPGAs werden meist für extreme Bedingungen verwendet (z.B. $4000 \ Op/\mu s$) !


Technologie

Layout




- Logic Cell: Mass zum Vergleich von FPGAs verschiedenere Familien/Hersteller
- Configurable Logic Block (CLB): Enthält 2 slices
- Slice: 4 Funktionsgeneratoren (4×LUT6) + 8 FFs
- **Distributed RAM**: Konfigurationsspeicher einiger Slices ist als Datenspeicher nutzbar
- Block RAM statische dual-port RAMs in Blöcken zu 18/36Kbit
- **DSP48 Slice**: Multiply-Accumulate und Arithmetic-Logic Unit (DSP ALU)
- **CMT**: Clock Management Tile (clock synthesis, phase shift, PLL)





Routing Ressourcen

Hierarchisches Routing mit verschiedenen langen Verbindungen ; Sechs-Transistor-Kreuzungspunkt ; Routing-Delay $\geq 70\%$ Gesamt-Delay ; Spezielles low-skew Netze für regionale/globale Clocksignale



Integrated Logic Analyzer (ILA) ———

ILAs werden meist für In-System-Debugging von FPGA-Designs verwendet (z.B. bei aufwendigen Schaltungen oder fehlenden Inputsignal-Spezifikationen).

- ILA werden zusammen mit dem Design-under-Test synthetisiert → genügend Ressourcen verfügbar
- ILA beeinflusst als Messmittel das Messobjekt → befolgen ILA & DuT die Regeln des synchronen Designs, ist dies akzeptabel.


$$T_{win} = N_S \cdot T_{CIk}$$

$$C_{SB} \geq N_S \cdot W_S$$

$$W_{S} = S + 1$$

1: immer da, evtl. Trigger

 T_{win} : beobachtbares Zeitfenster N_S : Anzahl zu speichernde Samples C_{SB} : Speicherkapazität des Sample Buffers W_S : Gesamtwortbreite aller zu anal. Signale S: Anzahl zu analysierende Signale



Beispiel


4 PWM Zyklen einer RGB-LED (Counter + Output) $\rightarrow N_R = 2$, $N_G = 3$, $N_B = 4$, $T_{DAC} = 100 \mu s$

$$W_S = \left. \begin{array}{cc} R: & 1+2 \\ G: & 1+3 \\ B: & 1+4 \end{array} \right\} = 12 + \mathbf{1} = 13$$

1: immer da, evtl. Trigger

$$T_{win} = 4 \cdot T_{DAC} = 400 \mu s$$

$$N_S \geq T_{win} \div T_{CLK} = 50000 \rightarrow 64K$$



Kurzanleitnung


- 1. Run Synthesis + Open Synthesized Design
- 2. Debug-Signale auswählen
 - 1. Netlist Window → Mark Debug
 - 2. Schematic Window → Mark Debug
 - Tcl-Console set_property MARK_DEBUG true [<signals>]
- 3. Signale mit Debug verbinden
 - Tools → Setup Debug
 - Tiefe Sample Buffer (N_S)
 - Capture Control + Advanced Trigger wählen
- 4. Save Constraints + existing Constraint File
- 5. Run Implementation + Generate Bitstream
 - 1. Target beschreiben
 - 2. ILA GUI öffnet sich

VHDL —

■ VHDL?

VHDL steht für **V**ery High Speed Integrated Circuit **H**ardware **D**escription **L**anguage, klingt schnell und ist es auch. Diese Sprache dient für die Hardwarebeschreibung von FPGAs, insbesondere wie die "Logikfläche" konfiguriert wird, und ist keine Programmiersprache, es handelt sich um eine Designsprache.

Einordnung VHDL als HDL



HW-nahes Design ermöglicht:

- + kleinere, schnellere, energie-effizientere Schaltungen als ESL-Design
- auf Kosten der Entwicklungszeit.

Entwicklung

Designflow

1. Spezifikation

 Erstellen/Verstehen von Funktions- und Testspezifikation. Vorgaben zum strukturellen Aufbau des Designs.

2. Architektur-Entwurf

- ullet Schaltung wird in Blockdiagramm festhalten o Ableiten von Ports, Wortbreiten, Codierung
- Je nach Komplexität Erstellung von Prozess-Dokumentation und/oder RTL-Schemas
- Zustandsdiagramm der FSMs

3. VHDL Implementierung

- Verwendung von VHDL-Templates für synchrone Logik
- VHDL-Code kommentieren

4. Design Constraints

• in .xdc-File *Top-Level Ports Location & Clock Period* Constraints setzen.

5. **Probe-Synthese**

 VHDL-Code überarbeiten bis keine Warnungen → Found latch for signal..., ... signals missing in the process sensitivity list..., ... signals form a combinational loop... Konsistenz-Check anhand Vergleich selbst gezählte #FF und Synthese-#FF

6. Simulation


- Erstellen einer VHDL-Testbench und Simulation des Designs (MUT) gemäss Spezifikation
- Testbench mit automatischem Vergleich von Ist & Soll ist für komplexere Designs gut

7. FPGA Implementierung

- Run Implementation ausführen für Technology-Optimizations & Place&Route
- ullet Falls Timinganalyse Fehler o Architektur überprüfen

8. HW-Test


• Bitstream auf HW gemäss Spezifikationen testen.




Simulation



Wie kann Parallelität auf einem sequentiell arbeitenden Computer simuliert werden? \rightarrow Event Queue & Transactions





Event Driven Simulation

- 1. Vorrücken der Simulationszeit zur nächsten Transaktion in der Event Queue.
- 2. Setzt alle zu aktualisierenden Signale auf den mit der aktuellen Transaktion verbundenen Wert.
- Arbeitet alle Prozesse ab, die auf Signale sensitiv sind, deren Wert sich durch die aktuelle Transaktion geändert hat. Signalzuweisungen werden als zukünftige Transaktionen in die Event Queue abgelegt.

Important

- Events werden durch die Signaländerung generiert, nicht die Transaktionen!
- Nicht alle Transaktionen führen zu einem Event (Sensitivity List)

Signale & Variablen

- ullet Variable Assignment := o unabhängig von Simulation, Effekt ist sofort
- Signal Assignment \leftarrow Effekt nach after (<u>nur</u> Simulation) oder bei keiner Verzögerung nach Simulationszyklus (δ -Delay ; Sim. & Synth)

- 1. Effekt nach after (nur Simulation)
- 2. Bei keiner Verzögerung (Synthese & Simulation) \rightarrow Transaction für nächsten Simulationszyklus geplant (δ -Delay)

Sprachelemente

Bezeichner

- z.B. Signaldeklaration oder Port-Variable
 - Gross-/kleinschreibung wird nicht unterschieden \rightarrow <u>nicht</u> case-sensitiv

```
signal some_signal, some_other_signal, result :

    integer;

ReSULT <= SOME_SIGNAL + sOME_oTHer_SIGNal;</pre>
```

- Namen können beliebig lang sein
- Keine Spezialzeichen ausser _ → nicht am Anfang & Ende + nicht verdoppelt

Signalzuweisung / Treiber <=

Mit dem *Treiber* <= werden Signale von der rechten Seite ausgewertet und auf das Signal auf der linken Seite zugewiesen.

Je nachdem wo <= verwendet wird, hat es eine andere Bedeutung \rightarrow in einem if-Statement wird es als kleinergleich angesehen ; bei Signalen als Treiber.

Kommentare

 $\mbox{Mit -- werden Kommentare begonnen} \rightarrow \mbox{single line comments!}$

```
-- Das hier ist ein Kommentar
Hier aber nicht mehr :(
```

Komponenten

.vhd-Dateistruktur

VHDL-Code wird meistens in .vhd Dateien geschrieben.

```
-- Header Commment (Author, Date, Filename, etc.)

library ... -- Library einbinden

use ... -- Packages aus Library bekanntgeben

entity ...

-- Schnittstelle der Komponente gegen aussen

architecture ...

-- Funktion (Innenleben) der Komponente
```

Entity

Eine Entity beschreibt den Komponenten für äusserliche Zugriffe \rightarrow nur Struktur der Komponente bekannt, aber nicht deren Inhalt.

```
entity MyComponent is
                                                    (1)
 generic(
   y : integer := 20;
   z : integer
 );
 port (
                                                    (2)
   a_pi, b_pi : in
                      std_logic; -- Input
         : out std_logic; -- Output
   --x_pio : inout std_logic -- Bidirectional
 );
 constant c_max_cnt : integer := 20_000;
                                                    (3)
end MyComponent;
```

- ① Analog zu #define in C \rightarrow werden während Kompilation eingefügt! "Ko"
- ② Signal Deklarationen → müssen bei Instanziierung des Komponenten im übergeordneten Design verdrahtet werden!

Sichtbarkeit

- Alles was in der Entity bekannt ist (inkl. Libraries), ist auch in der zugehörigen Architecture bekannt.
- Alles was in der Architecture bekannt ist, ist <u>nicht</u> in der Entity bekannt.

Generics

Analog zu C Preprocessor Direktive #define

Mit generic können Komponenten angepasst/parametrisiert werden \to Komponent muss daher mit diesen Generics implementiert werden.

Im Gegensatz zu Konstanten kann Wert <u>ausserhalb</u> definiert werden.

Beim Testen sollten mit Generic-Parametern die Randwerte verifiziert werden (*Corner-Case* Testing).

Beschreibung der Komponente

```
entity NAND_gate is
  generic(
    IW : integer := 2 -- input width, def. 2
);
  port(
    InP : in std_logic_vector(IW-1 downto 0);
    OutP : out std_logic;
);
end NAND_gate;
```

Verwendung der Komponente

```
architecture rtl of top is
component NAND_gate is
  generic(IW : integer := 3); -- default
  port(
    InP : in std_logic_vector(IW-1 downto 0);
    OutP : out std_logic;
 );
end component NAND_gate;
begin
I1 : NAND_gate
port map (
 InP \Rightarrow In1,
 OutP => 01
);
I2 : NAND_gate
generic map (
  IW => 8
             -- Instance 2 has 8 inputs
) -- note no semicolon
port map (
 InP \Rightarrow In1,
 OutP => 01
);
end logic;
```

Loop & Generate Statements

Bei Implementierung von Komponenten mit Generic-Parametern werden häufig loop oder generate verwendet.

```
-- architecture using loop statement
architecture A_loop of NAND_gate is
begin
  P_nand: process(InP)
    variable tmp : std_logic;
  begin
    tmp := InP(0);
    for i in 1 to IW-1 loop
      tmp := tmp and InP(i);
    end loop:
    OutP <= not tmp;
  end process;
end A_loop;
-- the same is achieved with generate statements
architecture A_gen of NAND_gate is
signal tmp : std_logic_vector(IW-1 downto 0);
begin
  tmp(0) \leq Inp(0);
  tree: for i in 1 to IW-1 generate
    tmp(i) \le tmp(i-1) and Inp(i);
```

```
c_po : out std_logic;
 );
end top;
architecture struct of top is
  signal a,b : std_logic;
 component MyComponent is
                                                        1
   port (
      a_pi, b_pi : in std_logic;
                 : out std_logic;
      c_po
   );
 end component MyComponent;
begin
 Inst1: MyComponent
                                                        2
 port map ( a_pi => a,
                                                        (3)
             b_pi \Rightarrow b,
             c_po => c_po
                                                        (4)
end struct;
```

- (1) Deklaration der Komponente im Deklarationsteil → Name der entsprechenden Entity!
- (2) Instanzierung eines Komponenten
- (3) Signale/Ports werden verbunden
- (4) Kein Endzeichen!

Es gibt allerdings eine Kurzschreibweise die angewendet werden kann, wenn sich die Komponente im selben Verzeichnis befindet

```
architecture struct of top is
  -- external module
 use work.MyComponent;
begin
 Inst1: entity work.MyComponent
    port map(
      a_pi => a,
      b_pi \Rightarrow b,
      c_po => c_po
    );
end struct;
```

Datentypen

Synthese & Simulation	Nur Simulation
std_logic, std_ulogic (IEEE 1164) signed, unsigned (IEEE 1176.3) integer (IEEE1076) type (IEEE 1076 userdefined)	real, time, character, file (IEEE 1076)

Integer

```
signal my_int : integer range -128 to 127;
```

- Repräsentiert Bereich $-2^{31} \dots 2^{31} 1$
- !!! Wertebereich einschränken (Konsistenzcheck für Simulation)
- Zahlen Darstellungen

```
-- Dezimal (Default)
my_int <= 61;
my_int <= 10#61#;
                      -- dezimal
my_int <= 2#111101#; -- binär
my_int <= 8#75#;
                       -- octal
my_int <= 16#3D#;</pre>
                       -- hexadezimal
```

Aufzählungstypen

Aufzählungstypen sind wie enum in C, wobei die Repräsentierung hinter den Aufzählunstypen nicht bekannt ist.



🛕 Reservierte Wörter

Für Aufzählungstypen dürfen keine reservierten Wörter verwendet werden. Im folgenden Beispiel wäre also port keine möglicher Wein, da dies ein reserviertes Wort ist.

```
type wine is (white, rose, red);
signal beverage : wine := red; -- initialization
```

Subtypes

gleiche Operationen wie Grundtype, einfach bestimmte Teilmenge (z.B. natural, positive)

```
subtype t_day is integer range 1 to 31;
signal day : t_day;
```

Array

Gruppierung von Elementen gleichen Types

```
type t_byte is array (7 downto 0) of std_logic;
signal byte : t_byte; -- same as below
signal byte : std_logic_vector(7 downto 0);
```

Attributes A'<atr>(N)

Mit den Attributes eines Arrays können verschiedene Informationen entnommen werden, welche während der Synthetisierung eingefügt werden (analog zu C Preprozessoren).

```
-- Initialise array(1. Dim, 2. Dim)
type t_arr is array(2 to 4, 15 downto 0) of

    std_logic;

signal A : t_arr;
A'left(1)
A'right(2)
                    -- 15
A'high(2)
A'low(2)
                   -- 2 to 4
A'range(1)
A'reverse_range(2) -- 0 to 15
                   -- 3
A'length(1)
A'ascending(2)
                    -- false
A'element
                    -- std_logic
```


Aggregates

Array Aggregates werden verwendet, um Arrays mit konstanten Werten einzusetzen.

Zuordnungen dürfen nicht gemischt werden!

```
-- !ERROR!
constant c_map : t_map := ('1', 2=>'0', others =>
$\to '1'$)
```

Die Indexierung von Arrays erfolgt analog zu C



Record

Gruppierung von Elementen unterschiedlichen Types

```
type t_date is record
  day : t_day;
  year : positive;
end record;
```

Signed/Unsigned (IEEE 1076.3)


Binärzahlen in Form von 2er-Komplement, bzw. vorzeichenlosen Binär-Arrays

Unterschied Signed/Unsigned und Integer

integer ist ein Skalar-Typ und fest in VHDL eingebaut. Obwohl der Typ der Zahlenbereich einer 32-Bit 2er-Komplementzahl hat, hat es keine Tool-interne Darstellung (z.B. MSB ist nicht prüfbar).

unsigned/signed sind definiert im Package numeric_std und haben eine genau definierte Darstellung, als Array von std_logic bits mit bekannten Definierung von MSB & LSB.

Umwandlungstabelle




IEEE std_logic_1164

Datentypen: std_logic & std_ulogic

i Warum std_logic_1164?

In Standard-VHDL gibt es zwei binäre Datentypen: bit (0,1) und boolean (false, true) \rightarrow entspricht aber nicht realen digitalen Signalen!



State	Bedeutung	Bereich
'U'	Uninitialized (–)	Simulation
'X'	Forcing Unknown (S1,S3)	Simulation
'0'	Forcing Low (S3)	Synthese, Simulation
'1'	Forcing High (S1)	Synthese, Simulation
'Z'	High Impedanz (–)	Synthese, Simulation
'W'	Weak Unknown ()	Simulation
'L'	Weak Low (S4)	Simulation
'H'	Weak High (S2)	Simulation
1_1	Don't Care	Simulation

– bedeutet alle Schalter offen

Caution

 std_ulogic steht für $unresolved \rightarrow Signal$ kann nur von einem Prozess geändert werden!

std_logic ist *resolved* und kann von mehreren Prozessen geändert werden. Zustände werden mit der *Resolution Table* aufgelöst.

Wird dies gemacht, Simulation möglich, aber keine Synthetisierung!

Real (Simulationstype)

 \rightarrow Wertebereich ist Hersteller-spezifisch

```
signal float : real;
float <= 73_000.0;
float <= 7.3E4;
float <= 73000;</pre>
①
```

(1) Fehler, da 730000 eine Ganzzahl ist

Time (Simulationstyp)

ightarrow einziger vordefinierter physikalischer Datentyp

```
type time is range -2147483647 to 2147483647
units fs;
   ps = 1000 fs;
   ns = 1000 ps;
   us = 1000 ns;
   ms = 1000 us;
   sec = 1000 ms;
   min = 60 sec;
   hr = 60 min;
end units;

signal t : time
t <= now + 2.5 sec;</pre>

10
```


(1) now liefert aktuelle Simulationszeit

Prozesse

Parallelität

In VHDL wird das Verhalten digitaler HW durch parallele Prozesse beschrieben die gleichzeitig ausgeführt werden und über Signale miteinander kommunizieren.

```
P1: process (i1, i2, i3) (1) variable v_tmp : std_logic; (2)
```



wait (nur für Simulation)

Mit dem Keyword wait kann ein Prozess pausiert und/oder sensitiv gemacht werden. **NUR FÜR SIMULATION!**

Nur Simulation

Simulation & Synthese

```
process
begin
    sigB <= sigA;
    wait on sigA;
end process;</pre>
process(sigA)
begin
    sigB <= sigA;
    end process;</pre>
```

Process Statement

Important

Ein Process Statement ist parallel nach aussen und sequentiell im Inneren.

Alle Signal Assignments ausserhalb von process (Concurrent-, Selected-, Conditional-Signal Assignment) sind Process Statements in Kurzschreibform!

Sequential Statements

Innerhalb eines Process sind alle Statements $\underline{\text{sequential state-}}$ ments

```
(sequential) signal assignment
variable assignment
if, case, for, loop, while
nur für Simulation:
wait, assert, report
```

Folgende beschreibungen resultieren in der selben implementierungen auf dem FPGA.

Schaltungssynthese

Synthese

- Synthese übersetzt VHDL-Modell in eine RTL-Beschreibung aus Registern und kombinatorischen Blöcken
- 2. Erzeugung von Netlist → Verbindungen von Elementen (Primitives), welche in den Slices verfügbar sind.

Implementation

- 1. **Initialize** \rightarrow Einbindung Constraints & Netlist ins Design
- Optimize → Packt Primitives der synth. Netlist in Slices, Versuch Gatterkomb. zu vereinfachen (Reduktion HW)
- 3. Place \rightarrow Zuweisung der Slices auf CLBs (configurable logic blocks) auf dem FPGA, Zuordnung IO aus Constraints
- 4. Route → Definiert Verbindungen zwischen CLB & IO
- Static Timing Analysis → Überprüfung des Designs mit den Timing Constraints.


! Synthesis vs. Implementation

- Synthesis generiert die Netlist des VHDL-Codes und beschreibt die Zusammensetzung
- *Implementation* wendet die Contraints an und sorgt für die Hardware-Implementierung

Bitstream

Die Bitstream-Generierung erzeugt das FPGA-Konfigurationsfile ('kompilierte Datei').

Speichermodellierung



Adressierbare ROM & RAM können im FPGA aufgebaut werden aus: Slice-Logik (LUTs, FF), speziellen Makrozellen (BRAM) und Konfigurations-RAM (Distributed Memory).

- RAMs im FPGA sind immer getaktete/synchrone Speicher
- ROMs können auch kombinatorisch sein falls keine Latenz besteht
- Nur kleine Speicher sollte man in behavioural VHDL beschreiben

kombinatorisch ROM

```
library IEEE;
use IEEE.STD_LOGIC_1164.all;
use IEEE.Numeric_Std.all;
entity crom is
 generic(
   AW : integer := 3; DW : integer := 8);
   addr : in
   std_logic_vector(AW-1 downto 0);
   Dout : out std_logic_vector(DW-1 downto 0)
 );
end entity crom;
architecture Behav of crom is
 type t_rom is array (0 to 2**AW-1) of
 std_logic_vector(DW-1 downto 0);
 constant rom : t_{rom} := (X"1A", X"1B", X"1C", X"1D",
                           X"2A", X"2B", X"2C", X"2D");
begin
 Dout <= rom(to_integer(unsigned(addr)));</pre>
end architecture Behav;
```

$$C = 2^{AW} \cdot DW = 2^{10} \cdot \underbrace{x}_{\text{[kBit]}}$$

C : Speicherkapazität in [Bits] oder [kBits]

AW : Adresswortbreite DW : Datenwortbreite

Beispiel ROM


$$C_{ROM} = 2^3 \cdot 8 = 64$$
 Bits $\rightarrow \#FF = 0$ (da asynchron)

Synchrones ROM "Write before Read"


WBR & RBW


Bei Write-Before-Read wird zuerst geschrieben und dann das Beschriebene gelesen.

Bei *Read-Before-Write* wird zuerst die Speicherstelle ausgelesen und erst dann die Stelle überschrieben.



Folgende RTL-Schema beschreiben ein WBR & RBW Speicher mit 1cc oder 0cc Latenz.





```
entity sram is
 generic(
   AW : integer := 4; DW : integer := 8);
  port (
   clk : in std_logic;
   we : in std_logic;
   addr : in std_logic_vector(AW-1 downto 0);
   Din : in std_logic_vector(DW-1 downto 0);
    Dout : out std_logic_vector(DW-1 downto 0)
 );
end entity sram;
architecture Behav of sram is
  type t_ram is array (0 to 2**AW-1) of
   std_logic_vector(DW-1 downto 0);
 signal ram : t_ram;
 signal r_addr : std_logic_vector(AW-1 downto 0);
 P_ram: process(clk)
 begin
   if rising_edge(clk) then
      if we = '1' then
        ram(to_integer(unsigned(addr))) <= Din;</pre>
      end if;
```

```
r_addr <= addr;
end if;
end process;

Dout <= ram(to_integer(unsigned(r_addr)));
end architecture Behav;</pre>
```

Beispiel RAM

$$C_{RAM} = 2^4 \cdot 8 = 128 \text{ Bits}$$


 $\rightarrow \#FF = \underbrace{4}_{AW} + \underbrace{128}_{Speicher} = 132 \text{ FF}$

Synchrone Logik -

i Warum synchrones Design?


synchrones Design: Eine worst-case Timing-Analyse

$$max(T_{Delay}) < T_{CLK} \pm T_{skew}$$




- + Keinen Einfluss von Hazards, ungültige Zwischenwerte, etc.
- + Signale vor Speicherung stabil
- + Deterministisches Verhalten unabhängig von Gate-level
- + Systematisches Design/Test/Debug mit etablierten Methoden & Tools
- Max. Verarbeitungsgeschwindigkeit durch Verzögerungszeit des längsten Pfades definiert.
- Evtl. höherer Energieverbrauch und EMV-Probleme durch CLK-Signal

asynchrones Design: ∞-viele Timing Analysen



Synchronisation & Entprellung

Metastabilität



→ Durch Verletzung der Hold-/Setup-Zeit kann ein Speicherelemnt in den Metastabilen Zustand geraten (unbestimmter Ausgang) ⇒ Kann durch Synchronisation reduziert werden!

Mean Time Between Failure

Zeit t_{meta res} beschreibt die Zeit, bis von der Metastabilität wieder ein definierten Wert angenommen wird.→ Je kleiner, desto besser!

$$t_{meta_res} < t_{allowed} = \frac{1}{f_{clk}} - t_{pd} - t_{su}$$

t_{MTBF} beschreibt die Wahrscheinlichkeit, dass die obere Bedingung nicht erfüllt ist \rightarrow Je kleiner desto besser!

$$t_{MTBF} = \frac{e^{K_2 \cdot \left(\frac{1}{f_{clk}} - t_{pd} - t_{su}\right)}}{K_1 \cdot f_{clk} \cdot f_d}$$

Grösstenteils ist die t_{MTBF} abhängig von der Clockfrequenz!

 K_1 : Prozess-Konstante [s] K_2 : Prozess-Konstante [Hz]

 t_{pd} : Propagation delay t_{su} : Setup-Zeit

: durchschnittliche Frequenz des asynchronen Datensignals

 f_{clk} : Clockfrequenz

Reset Synchronisierung

i Synchroner Reset


Synchronisierung wie für binäre Datensignale.

+ Geringer LUT-Verbrauch durch logische Kombination von Daten/Reset - Nur funktionstüchtig wenn Clock-Signal aktiv

i Asynchroner Reset

Spezielle Synchronisierung damit alle FFs im gleichen cc freigegeben werden.


+ Geringer LUT-Verbrauch - Höherer LUT-Verbrauch, keine logischen Kombinationen von Daten/Reset möglich



```
sync_rst: process(clk_pi, rst_pi)
begin
  if rst_pi = '1' then
    rst_sr <= (others => '1');
    rst_po <= '1';
  elsif rising_edge(clk_pi) then
    rst_sr(0)
                       <= '0';
    rst_sr(3 downto 1) <= rst_sr(2 downto 0);</pre>
    rst po
                        <= rst_sr(3);
  end if;
end process;
```


Entprellen

durch Blanking



- Kontaktsignal wird möglichst schnell abgetastet, um Signaländerungen auszuwerten.
- Geplante Aktion sofort beim Drücken bzw. Loslassen ausgeführt werden!
- Es muss T_{Blank} gewartet werden bis nächste Auswertung


durch Unterabtastung



- Kontaktsignal wird langsam abgetastet
- $T_{Sample} > T_{Prell}$

worst-case geplante Aktion erst zwei volle Abtastperioden nach Schalterereignis ausgeführt

Drehgeber-Signale (Quadratur-Signale)



Die Reihenfolge des Auftretens der 4 Eingangskombination bestimmt die aktuelle Drehrichtung. Eine sichere Decodierung ist z.B. mit einer Mealy-FSM mit 5 Zuständen möglich.

Finite State Machines (FSM)

Oder auch getaktete/synchrone/sequentielle Logik

i Warum FSM?

Jede (komplexe) digitale Schaltung benötigt ein "Gedächtnis" um Zustände zu speichern.

Eine Zustandsmaschine beschreibt ein System in diskreten Zuständen. In **VHDL** wird für Mealy- & Moore-Automaten jeweils ein <u>memoryless</u> und ein <u>memorizing</u> Prozess verwendet. Der <u>memoryless</u> Prozess verarbeitet die Zustandswechsel und die Ausgänge (wobei dies Abhängig vom FSM-Typ ist). Der <u>memorizing</u> Prozess ist für die Zustands-Zurücksetzung und -zuweisung zuständig.


i Allgemeine Definition ZSM

$$o[k] = g(i[k], s[k])$$

$$s[k+1] = f(i[k], s[k])$$

- diskrete Zeit mit $t = k \cdot T_{CLK}$, k = 0 entspricht Reset-Zeitpunkt
- Zustand des Systems mit
- $s \in S = \{S_0, S_1, \dots S_N\}$
- i: Input des Systems mit $i \in I = \{I_0, I_1, \dots I_M\}$
- Output des Systems mit
- $o \in O = \{O_0, O_1, \dots, O_K\}$
- Output Funktion, berechnet aktuellen Output des
- ⁹ · Systems
- $_{f}$. Next-State Funktion, berechnet nächsten
- Zustand des Systems

FSM-Typ: Mealy



$$o[k] = g(i[k], s[k])$$

 $s[k+1] = f(i[k], s[k])$

Beim *Mealy* werden die <u>Ausgänge</u> sowohl <u>vom aktuellen Zustand</u> als auch von den <u>aktuellen Eingängen bestimmt</u>. Es handelt sich daher um einen *0-delay-enable-type*




Abbildung 1: Mealy

$$o[k] = s[k]$$

$$s[k+1] = f(i[k], s[k])$$

Parasitäre Zustände

Jedes weitere Zustands-Flip-Flop erweitert die Anzahl Faktoren um den Faktor 2 ($S=2^N$). Ungebrauchte Zustände werden parasitäre Zustände genannt.

$$n_{para} = 2^N - S$$
 $n_{para}|_{S=3, N=2} = 2^2 - 3 = 1$

Folgende Formel kann die Anzahl benötigten Flip-Flops berech-

$$N = \lceil \log_2(S) \rceil = \left\lceil \frac{\ln(S)}{\ln(2)} \right\rceil$$
 $N|_{S=3} = \lceil \log_2(5) \rceil = 3$

N : Anzahl Bits (≘ Flip-Flops) S: Anzahl verwendete Zustände

State Encoding

Zustände können auf verschiedene Arten dargestellt werden, bekannte Varianten sind binär und One Hot.

Zustand	Binär	One-Hot		
S_0	00	001		
S_1	01	010		
S_2	10	100		
Parasitäre	11	000, 011, 111, 110,		
Zustände		101		



Parasitäre Zustände

Alle ungebrauchten Zustände sind parasitäre Zustände! Parasitäre Zustände sind umbedingt abzufangen:

Binär

Meistverwendetes Format ist binär, da es kompakt und einfach erweiterbar ist.

- $S_0 \rightarrow 0000$
- $S_1 \rightarrow 0001$

• $S_2 \rightarrow 0010$

One-Hot

Bei One-Hot ist ein Bit high und alle anderen Bits low oder in anderen Worten, nur ein Bit ist aktiv.

Goldene Regeln der (FSM) Implementierung

Memoryless Process (kombinatorische Logik)

- Alle Eingangssignale der FSM und der aktuelle Zustand müssen in der sensitivity list aufgeführt werden.
- Jedem Ausgangssignal muss für jede mögliche Kombination von Eingangswerten (inkl. parasitäre Input-Symbole) ein Wert zugewiesen werden. Keine Zuweisung bedeutet sequentielles Verhalten (Speicher)!
- Parasitäre Zustände sollten mittels others abgefangen werden.

Memorizing Process (sequentielle Logik)

- Ausser Clock und (asynchronem) Reset dürfen keine Signale in die sensitivity list aufgenommen werden.
- Das den Zustand repräsentierende Signal muss einen Reset-Wert erhalten.




Latches

Latch-Warnungen bei der Synthese deuten gut auf eine Missachtung der Regeln.

Fest-/ und Gleitkomma-Arithmetik —

Festkomma



$$W \ge F \ge 0$$

W : Gesamtgrösse [Bit] F: Nachkomma-Bits [Bit]

Vorzeichenlose Zahlen (unsigned) im Bereich

$$0 \le \sum_{k=0}^{W-1} b_k \cdot 2^{k-F} \le 2^{W-F} - 2^{-F}$$

! Zahlen mit Vorzeichen (signed) im Bereich

$$\begin{array}{l} -2^{W-F-1} \leq \\ -b_{W-1} \cdot 2^{W-F-1} + \sum_{k=0}^{W-2} b_k \cdot 2^{k-F} \leq \\ 2^{W-F-1} - 2^{-F} \end{array}$$

Die Auflösung R ist im Festkomma-Format im Gegensatz zu Gleitkomma über den gesamten Wertebereich konstant. Es beschreibt wie gross ein Bit ist.

$$R = 2^{-F}$$
 \rightarrow $R|_{F=8} = 2^{-8} \approx 0.00391$

Bereich ausrechnen

unsigned

$$min = 0 ; max = R \cdot (2^W - 1)$$

Range =
$$[0, R \cdot (2^W)]$$

Beispiel W = 4, F = 3

$$\begin{array}{ll} \text{min:} & \text{0.000} = 0 \\ \text{max:} & \text{1.111} = 1.875 \end{array} \right\} \Rightarrow [0, 2)$$

signed


$$\min = -R \cdot (2^{W-1}) : \max = R \cdot (2^{W-1} - 1)$$

Range =
$$[-R \cdot 2^{W-1}, R \cdot 2^{W-1})$$

Beispiel W = 4, F = 3

$$\begin{array}{ll} \text{min:} & \text{1.000} = -1 \\ \text{max:} & \text{0.111} = \text{0.875} \end{array} \right\} \Rightarrow [-1, \, 1)$$

Addition




Sign-Extension bei A wird verwendet, falls das Carry-Bit benötigt wird.

- (1) Sign-Extension = 0 falls unsigned Arithmetik
- 2 Sign-Extension = MSB falls signed Arithmetik

$$W_S = \max(F_A, F_B) + \max(W_A - F_A, W_B - F_B) + 1$$

$$F_S = \max(F_A, F_B)$$

Multiplikation



Kein Zero-Padding oder Sign-Extension nötig. Das Produkt hat stets genauso viele *gedachte* Vor-/Nachkomma-Stelle wie die Summe der Vor-/Nachkomma-Stellen beider Operanden.

$$P'length = A'length + B'length$$

$$W_P = W_A + W_B$$

$$F_P = F_A + F_B$$

Division vermeiden

Eine Multiplikation verläuft schneller als eine Division und sollte in allen möglichen Fällen vermieden werden!

Gleitkomma



Zusätzlich werden E Bits von W Bits verwendet, um die Lage des Binärpunktes zu kodieren und M=W-E Bits für die Auflösung verwendet.

$$D = (-1)^{s} \cdot \underbrace{(1 + m \cdot 2^{-M})}_{1 + x^{-1} + x^{-2} + x^{-3} + \cdots} \cdot 2^{(e-b)}$$

$$b = 2^{E-1} - 1$$

s: Vorzeichebit (Sign)

m : vorzeichenloser Wert der Mantisse M

e : vorzeichenloser Wert des Exponenten E

b: Wert des Exponenten-Bias $b = 2^{E-1} - 1$

Beispiel von oben:

$$b = 2^{E-1} - 1 = 2^{8-1} - 1 = 127$$

$$\underline{D} = (-1)^{1} \cdot (1 + \cdot 2^{-3} + 2^{-8} + 2^{-10}) \cdot 2^{131 - 127}$$
$$= -18.078125$$

IEEE-754	Name	W	Ε	М	min / max
2008	binary16 (half)	16	5	10	10 ^{±5}
1985	binary32 (single)	32	8	23	$10^{\pm 38}$
1985	binary64 (double)	64	11	52	$10^{\pm 308}$
2008	binary128 (quad)	128	15	112	$10^{\pm 4932}$

♦ Fest- und Gleitkomma

Das Festkomma-Format (FK) kann mit einer gegebenen Anzahl ${\cal W}$ Bits

- die Auflösung (absoluter Fehler) mit F = W optimieren
 - .XXXX
- ullet den darstellbaren Wertebereich mit F=0 optimieren
 - XXXX.
- aber nicht beides gleichzeitig! ⇒ Daher Gleitkomma

i Auflösung und Fehler

Fix-Point:

- Auflösung über Wertebereich gleich
- Fehler verändert sich


Float-Point:

- Fehler über Wertebereich gleich
- Auflösung verändert sich

PWM-D/A —

Mit dem PWM-Verfahren (+TP-Filter) kann aus einem Digitalwert ein Analogsignal erzeugt werden. Ein Referenzcounter dimensioniert auf *N*-Bits zählt hoch und ab einem Schwellwert wird das PWM-Signal von High auf Low gezogen:

- 1. Ref-Counter $n_{cnt} < d_{in} \rightarrow \text{pwm_out high}$
- 2. Ref-Counter $n_{cnt} \geq d_{in} \rightarrow \text{pwm_out } \overline{\text{low}}$




$$f_{DAC} = \frac{f_{CNT}}{2^N - 1} = \frac{f_{CLK}}{P \cdot (2^N - 1)}$$
$$D = \frac{d_{in}}{2^N - 1} \cdot 100\%$$

 f_{DAC} : Frequenz des PWM-Signals f_{CNT} : Frequenz einzelnes PWM-Bit

P: PWM-Prescaler

N: PWM-Auflösung in Bits

D: Tastgrad d_{in} : Inputwert



Packages -

In Packages sammelt man Deklarationen, die an mehreren Orten verwendet werden. Selbstdefinierte Typen, welche in Ports verwendet werden, **müssen** in Packages definiert werden. Packages mit Subprogramms **erfordern** immer eine Implementation.

```
-- Package Declaration
package my_pkg is
                                                      1
  type mix_rec is record
   element1: std_logic;
   element2: natural;
  end record;
  constant const_1: natural := 7;
  function f1(a,b:mix_rec) return std_logic;
end my_pkg;
-- Package Implementation
package body my_pkg is
                                                      2
  function f1(a,b:mix_rec) return std_logic is
                                                      (3)
    return a.element1 or b.element1;
  end f1;
```

```
end my_pkg;
```

- (1) Package Deklaration
- (2) Package Implementation
- (3) Implementation der Funktion f1

Libraries & Use Clause

- Design Units (Entity, Architecture, Package) sind in Libraries organisiert.
- Default-Bibliothek ist work → Eigene Bibliotheken in Vivado kann über Source File Properties des gewünschten Packages eingestellt/erstellt werden
- Deklarationen können auf zwei Arten zugegriffen werden

```
library myLib;
                                                        (1)
use myLib.my_pkg.mix_rec;
                                                        2
use myLib.my_pkg.all;
                                                        (3)
entity E1 is
                                                        (4)
 port(
   I : in myLib.my_pkg.mix_rec;
    0 : out myLib.my_pkg.mix_rec);
end E1;
entity E2 is
                                                        (5)
 port(
   I : in mix_rec;
    0 : out mix_rec);
end E2;
```

- (1) nicht work-Libraries müssen mit library geladen werden!
- (2) laden von spezifischen Deklarationen
- (3) alle Deklarationen eines Packages laden
- (4) Zugriff ohne use (direkt, ohne $\langle 2 \rangle \& \langle 3 \rangle$)
- (5) Zugriff mit use (angenehm)

Liste von Packages

Synthetisierbare Bibliotheken

ieee.std_logic_1164

```
library ieee;
use ieee.std_logic_1164.all;

library ieee;
use ieee.numeric_std.all;
```

Nicht-Synthetisierbare Bibliotheken

```
library ieee;
use ieee.math_real.all;
```

Vorlagen

Der Inhalt der Prozess-Templates wird in den =>CUSTOM gekennzeichneten Abschnitten geschrieben.

Positive Getriggertes D-FlipFlop

Mit asynchronem Reset



```
process (rst, clk) -- !!!


begin
   if rst = '1' then
     Q <= '0';
   elsif rising_edge(clk) then
     Q <= D;
   end if;
end process;</pre>

①

①
```

- (1) Deklarationen
- ② Asynchroner Reset
- (3) Getaktete Logik

Ohne Reset



```
process (clk) -- !!!


begin
  if rising_edge(clk) then
   Q <= D;
  end if;
end process;</pre>

①
```

- Deklarationen
- ② Getaktete Logik

Finite State Machine

Mealy




```
-- FSM initialization
type state is (S0, S1, S2);
signal c_st, n_st : state;
-- memorizing process
p_seq: process (rst, clk)
                                                        (1)
begin
 if rst = '1' then
   c_st <= S0;
 elsif rising_edge(clk) then
   c_st <= n_st;</pre>
 end if;
end process;
-- memoryless process
                                                        2
p_com: process (i, c_st)
begin
  -- default assignments
 n_st <= c_st; -- remain in current state</pre>
 o <= '1'; -- most frequent value
 -- specific assignments
 case c_st is
    when S0 =>
     if i = "00" then
        o <= '0';
        n_st <= S1;
     end if:
    when S1 =>
     if i = "00" then
        n_st <= S2;
      elsif i = "10" then
        n_st <= S0;
```

```
end if;
when S2 =>
    if i = "10" then
        o <= '0';
        n_st <= S0;
    elsif i = "11" then
        n_st <= S1;
    end if;
when others =>
    -- handle parasitic states
    n_st <= S0;
end case;
end process;</pre>
```

- Memorizing (sequentielle Logik)
- ② Memoryless (kombinatorische Logik)

Moore



```
-- FSM initialization
type state is (S0, S1, S2);
signal c_st, n_st : state;
```

```
-- memorizing process

p_seq: process (rst, clk)

begin

if rst = '1' then

    c_st <= 50;

elsif rising_edge(clk) then

    c_st <= n_st;

end if;

end process;

-- memoryless process

p_com: process (i, c_st)

begin


-- default assignments
```

```
n_st <= c_st; -- remain in current state</pre>
 o <= '1'; -- most frequent value
  -- specific assignments
 case c_st is
   when S0 =>
      if i = "00" then
       n_st <= S1;
     end if;
    when S1 =>
     if i = "00" then
        n_st <= S2;
     elsif i = "10" then
       n_st <= S0;
     end if;
     o <= '0'; -- uncondit. output assignment
    when S2 =>
     if i = "10" then
       n_st <= S0;
     elsif i = "11" then
        n_st <= S1;
     end if;
    when others =>
      -- handle parasitic states
     n_st <= S0;
 end case;
end process;
```

- Memorizing (sequentielle Logik)
- 2 Memoryless (kombinatorische Logik)

Synchronisation


einfach



```
-- initialize sync signal
signal sync: std_logic_vector(1 downto 0);

-- synchronisation
process (rst, clk)
begin
   if rst = '1' then
       sync <= "00";
   elsif rising_edge(clk) then
       sync(0) <= async;
       sync(1) <= sync(0);
   end if;
end process;</pre>
```

mit Flankenerkennung




```
-- initialize sync signal
signal sync: std_logic_vector(2 downto 0);

-- edge detection
enb <= sync(1) and not sync(2);

-- synchronisation
process (rst, clk)
begin
   if rst = '1' then
       sync <= "000";
   elsif rising_edge(clk) then
       sync(0) <= async;
       sync(1) <= sync(0);
       sync(2) <= sync(1);
   end if;
end process;</pre>
```

Entprellen


durch Blanking & Unterabtastung



```
deb_sig : process(clk)
begin
  if rising_edge(clk) then
   if deb_cnt = 0 then
    if (sig /= debncd_sig) then
        deb_cnt <= c_blank_time;
        debncd_sig <= sig;
   end if;
elsif deb_cnt > 0 then
   deb_cnt <= deb_cnt -1;
end if;
end if;</pre>
```

end process;

durch Unterabtastung



```
deb_sig : process(clk)
begin
  if rising_edge(clk) then
    if deb_cnt < c_sample_time then
        deb_cnt <= deb_cnt + 1;
    else
        debncd_sig <= sig;
        deb_cnt <= (others => '0');
    end if;
end process;
```







STOP DOING FPGAs

- CHIPS WERE NOT SUPPOSED TO BE REPROGRAMMABLE
- YEARS OF EXPERIMENTING yet NO REAL-WORLD USE FOUND for changing the chip's layout
- Wanted to play with logic gates? We had a tool for that: It was called "Logisim"
- "Yes please give me LUTs of something. Please give me 1GHz bus of it" - Statements dreamed up by evil wizards

LOOK at what FPGA addicts have been demanding your Respect for all this time, with all the programs & tools we built for them

(This is REAL circuitry, done by REAL FPGA designers):







"Hello I would like please"

They have played us for absolute fools